

# Machine-Level Programming I: Basics

COMP402127: Introduction to Computer Systems

<https://xjtu-ics.github.io/>

Danfeng Shan

Xi'an Jiaotong University

# Today: Machine Programming I: Basics




- **History of Intel processors and architectures**
- **Assembly Basics: Registers, operands, move**
- **Arithmetic & logical operations**
- **C, assembly, machine code**

# Intel x86 Processors

- **Dominate laptop/desktop/server market**
- **Evolutionary design**
  - Backwards compatible up until 8086, introduced in 1978
  - Added more features as time goes on
    - Now 3 volumes, about 5,000 pages of documentation
- **x86 is a Complex Instruction Set Computer (CISC)**
  - Many different instructions with many different formats (next slide)
  - *But!* only small subset encountered with most Linux programs
- **Compare: Reduced Instruction Set Computer (RISC)**
  - RISC: *\*very few\** instructions, with *\*very few\** modes for each
  - RISC can be quite fast (but Intel still wins on speed!)
  - Current RISC renaissance (e.g., ARM, RISC-V), especially for low-power

# Intel x86 Evolution: Milestones


<b>Name</b>	<b>Date</b>	<b>Transistors</b>	<b>MHz</b>
<b>8086</b> <ul style="list-style-type: none"> <li>First 16-bit Intel processor. Basis for IBM PC &amp; DOS</li> <li>1MB address space</li> </ul>	<b>1978</b>	<b>29K</b>	<b>5-10</b>
<b>386</b> <ul style="list-style-type: none"> <li>First 32-bit Intel processor (IA32). "Flat addressing"; runs Unix.</li> </ul>	<b>1985</b>	<b>275K</b>	<b>16-33</b>
<b>Pentium 4E</b> <ul style="list-style-type: none"> <li>First 64-bit Intel x86 processor (x86-64).</li> </ul>	<b>2004</b>	<b>125M</b>	<b>2800-3800</b>
<b>Core 2</b> <ul style="list-style-type: none"> <li>First multicore Intel processor</li> </ul>	<b>2006</b>	<b>291M</b>	<b>1060-3333</b>
<b>Core i7</b> <ul style="list-style-type: none"> <li>Four cores</li> </ul>	<b>2008</b>	<b>731M</b>	<b>1600-4400</b>



# Intel x86 Processors, cont.

## Machine Evolution

■ 386	1985	0.3M
■ Pentium	1993	3.1M
■ Pentium/MMX	1997	4.5M
■ PentiumPro	1995	6.5M
■ Pentium III	1999	8.2M
■ Pentium 4	2000	42M
■ Core 2 Duo	2006	291M
■ Core i7	2008	731M
■ Core i7 Skylake	2015	1.9B
■ Core i9 Raptor Lake	2023	25.9B



## Added Features

- Instructions to support multimedia operations, efficient conditionals
- Transition from 32 bits to 64 bits
- More cores

# Intel x86 Processors


## ■ Past Generations

		Process tech.
■ 1 <sup>st</sup> Pentium Pro	1995	600 nm
■ 1 <sup>st</sup> Pentium III	1999	250 nm
■ 1 <sup>st</sup> Pentium 4	2000	180 nm
■ 1 <sup>st</sup> Core 2 Duo	2006	65 nm


## ■ Recent & Upcoming Generations

1.	Nehalem	2008	45 nm
2.	Sandy Bridge	2011	32 nm
3.	Ivy Bridge	2012	22 nm
4.	Haswell	2013	22 nm
5.	Broadwell	2014	14 nm
6.	Skylake	2015	14 nm
7.	Kaby Lake	2016	14 nm
8.	Coffee Lake	2017	14 nm
9.	Cannon Lake	2018	10 nm
10.	Ice Lake	2019	10 nm
11.	Tiger Lake	2020	10 nm
12.	Alder Lake	2021	"intel 7" (10nm+++)
13.	Raptor Lake	2022	"intel 7" (10nm+++)
14.	Meteor Lake	2023	"Intel 4" (7nm)
15.	Arrow/Lunar Lake	2024	TSMC N3B (3nm)
16.	Panther Lake	2025	Intel 18A (1.8nm)

Process technology dimension  
= width of narrowest wires  
(10 nm  $\approx$  100 atoms wide)



Core i9-13900K






# Intel's Raptor Lake (2023)



In recent years, increasing die space devoted to the graphics/AI engine

# Intel's Latest: Arrow Lake (2024, Intel Ultra 9 285K )



P-Core

CPU Tile



# x86 Clones: Advanced Micro Devices (AMD)

## ■ Historically

- AMD had followed just behind Intel
- A little bit slower, a lot cheaper

## ■ Then

- Recruited top circuit designers from Digital Equipment Corp. and other downward trending companies
- Built Opteron: tough competitor to Pentium 4
- Developed x86-64, their own extension to 64 bits

## ■ Recent Years

- Intel got its act together
  - 1995-2011: Lead semiconductor “fab” in world
- 2015: TSMC becomes leading semiconductor fab; Intel falls behind
- Non-x86 GPUs from Nvidia now dominate compute market

# Intel's 64-Bit History

- **2001: Intel Attempts Radical Shift from IA32 to IA64**
  - Totally different architecture (Itanium)
  - Executes IA32 code only as legacy
  - Performance disappointing
- **2003: AMD Steps in with Evolutionary Solution**
  - x86-64 (now called "AMD64")
- **Intel Felt Obligated to Focus on IA64**
  - Hard to admit mistake or that AMD is better
- **2004: Intel Announces EM64T extension to IA32**
  - Extended Memory 64-bit Technology
  - Almost identical to x86-64!
- **All but low-end x86 processors support x86-64**
  - But, lots of code still runs in 32-bit mode

# Our Coverage

## ■ IA32

- The traditional x86
- <http://csapp.cs.cmu.edu/3e/waside.html>

## ■ x86-64

- The standard
- `linux> gcc hello.c`
- `linux> gcc -m64 hello.c`

## ■ Presentation

- Book covers x86-64
- Web aside on IA32
- We will only cover x86-64

# Today: Machine Programming I: Basics


- History of Intel processors and architectures
- **Assembly Basics: Registers, operands, move**
- Arithmetic & logical operations
- C, assembly, machine code

# Levels of Abstraction


## C programmer

```
#include <stdio.h>
int main(){
    int i, n = 10, t1 = 0, t2 = 1, nxt;
    for (i = 1; i <= n; ++i){
        printf("%d, ", t1);
        nxt = t1 + t2;
        t1 = t2;
        t2 = nxt; }
    return 0; }
```


## Assembly programmer



## Computer Designer




Gates, clocks, circuit layout, ...



# Definitions

- **Architecture:** (also ISA: instruction set architecture) The parts of a processor design that one needs to understand for writing assembly/machine code.
  - Examples: instruction set specification, registers
- **Microarchitecture: Implementation of the architecture**
  - Examples: cache sizes and core frequency
- **Code Forms:**
  - **Machine Code:** The byte-level programs that a processor executes
  - **Assembly Code:** A text representation of machine code
- **Example ISAs:**
  - Intel: x86, IA32, Itanium, x86-64
  - ARM: Used in almost all mobile phones
  - RISC V: New open-source ISA

# Assembly/Machine Code View



## Programmer-Visible State

- **PC: Program counter**
  - Address of next instruction
  - Called “RIP” (x86-64)
- **Register file**
  - Heavily used program data
- **Condition codes**
  - Store status information about most recent arithmetic or logical operation
  - Used for conditional branching
- **Memory**
  - Byte addressable array
  - Code and user data
  - Stack to support procedures

# Assembly: Data Types


- **“Integer” data of 1, 2, 4, or 8 bytes**
  - Data values
  - Addresses (untyped pointers)
  
- **Floating point data of 4, 8, or 10 bytes**
  
- **Code: Byte sequences encoding series of instructions**
  
- **No aggregate types such as arrays or structures**
  - Just contiguously allocated bytes in memory



# Assembly: Operations

- **Transfer data between memory and register**
  - Load data from memory into register
  - Store register data into memory
  
- **Perform arithmetic function on register or memory data**
  
- **Transfer control**
  - Unconditional jumps to/from procedures
  - Conditional branches
  - Indirect branches

Register names



```
addq %rbx, %rax
```

is

```
rax += rbx
```

**These are 64-bit registers, so we know this is a 64-bit add**

# Today: Machine Programming I: Basics

- History of Intel processors and architectures
- **Assembly Basics: Registers, operands, move**
- Arithmetic & logical operations
- C, assembly, machine code

# x86-64 Integer Registers

<b>%rax</b>	<b>%eax</b>
<b>%rbx</b>	<b>%ebx</b>
<b>%rcx</b>	<b>%ecx</b>
<b>%rdx</b>	<b>%edx</b>
<b>%rsi</b>	<b>%esi</b>
<b>%rdi</b>	<b>%edi</b>
<b>%rsp</b>	<b>%esp</b>
<b>%rbp</b>	<b>%ebp</b>

<b>%r8</b>	<b>%r8d</b>
<b>%r9</b>	<b>%r9d</b>
<b>%r10</b>	<b>%r10d</b>
<b>%r11</b>	<b>%r11d</b>
<b>%r12</b>	<b>%r12d</b>
<b>%r13</b>	<b>%r13d</b>
<b>%r14</b>	<b>%r14d</b>
<b>%r15</b>	<b>%r15d</b>

- Can reference low-order 4 bytes (also low-order 1 & 2 bytes)
- Not part of memory (or cache)

# Some History: IA32 Registers

			Origin (mostly obsolete)	
general purpose	<b>%eax</b>	<b>%ax</b>	<b>%ah</b>   <b>%al</b>	<i>accumulate</i>
	<b>%ecx</b>	<b>%cx</b>	<b>%ch</b>   <b>%cl</b>	<i>counter</i>
	<b>%edx</b>	<b>%dx</b>	<b>%dh</b>   <b>%dl</b>	<i>data</i>
	<b>%ebx</b>	<b>%bx</b>	<b>%bh</b>   <b>%bl</b>	<i>base</i>
	<b>%esi</b>	<b>%si</b>		<i>source index</i>
	<b>%edi</b>	<b>%di</b>		<i>destination index</i>
	<b>%esp</b>	<b>%sp</b>		<i>stack pointer</i>
	<b>%ebp</b>	<b>%bp</b>		<i>base pointer</i>

16-bit virtual registers  
(backwards compatibility)

**Most x64 instructions are two-operand:**

**movq src, dst**

**move "quad-word" (64 bits) from src to dst**

**Ex: movq %rax, %rcx** moves the value from register rax to rcx

**addq src, dst**

**Implements  $dst += src$**

**subq src, dst**

**Implements  $dst -= src$**

# Moving Data

## ■ Moving Data

`movq Source, Dest`

## ■ Operand Types

- **Immediate:** Constant integer data
  - Example: `$0x400`, `$-533`
  - Like C constant, but prefixed with `'$'`
  - Encoded with 1, 2, or 4 bytes
- **Register:** One of 16 integer registers
  - Example: `%rax`, `%r13`
  - But `%rsp` reserved for special use
  - Others have special uses for particular instructions
- **Memory** 8 consecutive bytes of memory at address given by register
  - Simplest example: `(%rax)`
  - Various other “addressing modes”

`%rax`

`%rcx`

`%rdx`

`%rbx`

`%rsi`

`%rdi`

`%rsp`

`%rbp`

`%rN`

**Warning: Intel docs use  
`mov Dest, Source`**

# BTW: Intel and AT&T Syntax

## ■ Intel Syntax (aka, MASM, Microsoft Macro Assembler)

- Microsoft (Visual Studio)

## ■ AT&T Syntax (aka, GAS, Gnu ASembler)

- GNU/GCC
- Default format for objdump (GNU Binutils)

## ■ Differences

### ■ Prefixes

- \$ vs. None

Intex Syntax	AT&T Syntax
mov eax, 1	movl \$1, %eax
mov ebx, 0ffh	movl \$0xff, %ebx
int 80h	int \$0x80

### ■ Direction of Operands

Intex Syntax	AT&T Syntax
instr dest, source	instr source, dest
mov eax, [ecx]	movl (%ecx), %eax

### ■ Memory Operands

- [] vs. ()

Intex Syntax	AT&T Syntax
mov eax, [ebx]	movl (%ebx), %eax
mov eax, [ebx+3]	movl 3(%ebx), %eax



# movq Operand Combinations

	Source	Dest	Src, Dest	C Analog
movq	Imm	Reg	movq \$0x4, %rax	temp = 0x4;
		Mem	movq \$-147, (%rax)	*p = -147;
	Reg	Reg	movq %rax, %rdx	temp2 = temp1;
		Mem	movq %rax, (%rdx)	*p = temp;
	Mem	Reg	movq (%rax), %rdx	temp = *p;

**Cannot do memory-memory transfer with a single instruction**

# Simple Memory Addressing Modes

## ■ Normal (R) Mem[Reg[R]]

- Register R specifies memory address
- Aha! Pointer dereferencing in C

```
movq (%rcx), %rax
```

# Example of Simple Addressing Modes

```
void swap
(long *xp, long *yp)
{
    long t0 = *xp;
    long t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

```
swap:
    movq    (%rdi), %rax
    movq    (%rsi), %rdx
    movq    %rdx, (%rdi)
    movq    %rax, (%rsi)
    ret
```


# Understanding Swap()

```
void swap
(long *xp, long *yp)
{
    long t0 = *xp;
    long t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

Register	Value
%rdi	xp
%rsi	yp
%rax	t0
%rdx	t1

swap:

```
movq    (%rdi), %rax    # t0 = *xp
movq    (%rsi), %rdx    # t1 = *yp
movq    %rdx, (%rdi)   # *xp = t1
movq    %rax, (%rsi)   # *yp = t0
ret
```



# Understanding Swap()

## Registers

<code>%rdi</code>	<code>0x120</code>
<code>%rsi</code>	<code>0x100</code>
<code>%rax</code>	
<code>%rdx</code>	

## Memory

	Address
123	<code>0x120</code>
	<code>0x118</code>
	<code>0x110</code>
	<code>0x108</code>
456	<code>0x100</code>


**swap:**

```

movq    (%rdi), %rax    # t0 = *xp
movq    (%rsi), %rdx    # t1 = *yp
movq    %rdx, (%rdi)    # *xp = t1
movq    %rax, (%rsi)    # *yp = t0
ret

```

# Understanding Swap()




swap:

```

movq    (%rdi), %rax    # t0 = *xp
movq    (%rsi), %rdx    # t1 = *yp
movq    %rdx, (%rdi)    # *xp = t1
movq    %rax, (%rsi)    # *yp = t0
ret

```

# Understanding Swap()




**swap:**

```

movq    (%rdi), %rax    # t0 = *xp
movq    (%rsi), %rdx    # t1 = *yp
movq    %rdx, (%rdi)    # *xp = t1
movq    %rax, (%rsi)    # *yp = t0
ret

```

# Understanding Swap()



**swap:**


```

movq    (%rdi), %rax    # t0 = *xp
movq    (%rsi), %rdx    # t1 = *yp
movq    %rdx, (%rdi)    # *xp = t1
movq    %rax, (%rsi)    # *yp = t0
ret

```



# Understanding Swap()



**swap:**

```

movq    (%rdi), %rax    # t0 = *xp
movq    (%rsi), %rdx    # t1 = *yp
movq    %rdx, (%rdi)    # *xp = t1
movq    %rax, (%rsi)    # *yp = t0
ret

```

# Simple Memory Addressing Modes

## ■ Normal (R) Mem[Reg[R]]

- Register R specifies memory address
- Aha! Pointer dereferencing in C

```
movq (%rcx), %rax
```

## ■ Displacement D(R) Mem[Reg[R]+D]

- Register R specifies start of memory region
- Constant displacement D specifies offset

```
movq 8(%rbp), %rdx
```

# Complete Memory Addressing Modes

## ■ Most General Form

**$D(Rb, Ri, S)$                        $Mem[Reg[Rb]+S*Reg[Ri]+ D]$**

- D:     Constant “displacement” 1, 2, or 4 bytes
- Rb:   Base register: Any of 16 integer registers
- Ri:    Index register: Any, except for `%rsp`
- S:     Scale: 1, 2, 4, or 8 (*why these numbers?*)

## ■ Special Cases

**$(Rb, Ri)$                                $Mem[Reg[Rb]+Reg[Ri]]$**

**$D(Rb, Ri)$                              $Mem[Reg[Rb]+Reg[Ri]+D]$**

**$(Rb, Ri, S)$                           $Mem[Reg[Rb]+S*Reg[Ri]]$**

# Address Computation Examples

<code>%rdx</code>	<code>0xf000</code>
<code>%rcx</code>	<code>0x0100</code>

## ■ Most General Form

- $D(Rb, Ri, S)$**        **$Mem[Reg[Rb]+S*Reg[Ri]+ D]$**
- D: Constant "displacement" 1, 2, or 4 bytes
  - Rb: Base register: Any of 16 integer registers
  - Ri: Index register: Any, except for `%rsp`
  - S: Scale: 1, 2, 4, or 8 (*why these numbers?*)

## ■ Special Cases

- $(Rb, Ri)$**        **$Mem[Reg[Rb]+Reg[Ri]]$**   
 **$D(Rb, Ri)$**        **$Mem[Reg[Rb]+Reg[Ri]+D]$**   
 **$(Rb, Ri, S)$**        **$Mem[Reg[Rb]+S*Reg[Ri]]$**

Expression	Address Computation	Address
<code>0x8(%rdx)</code>		
<code>(%rdx,%rcx)</code>		
<code>(%rdx,%rcx,4)</code>		
<code>0x80(,%rdx,2)</code>		

# Today: Machine Programming I: Basics

- History of Intel processors and architectures
- Assembly Basics: Registers, operands, move
- **Arithmetic & logical operations**
- C, assembly, machine code

# Address Computation Instruction

## ■ `leaq Src, Dst`

- Src is address mode expression
- Set Dst to address denoted by expression

## ■ Uses

- Computing addresses without a memory reference
  - E.g., translation of `p = &x[i];`
- Computing arithmetic expressions of the form  $x + k*y$ 
  - $k = 1, 2, 4, \text{ or } 8$

## ■ Example

```
long m12(long x)
{
    return x*12;
}
```

## Converted to ASM by compiler:

```
leaq (%rdi,%rdi,2), %rax # t = x+2*x
salq $2, %rax           # return t<<2
```

# Some Arithmetic Operations

## ■ Two Operand Instructions:

### Format

### Computation

<code>addq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} + \text{Src}$
<code>subq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} - \text{Src}$
<code>imulq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} * \text{Src}$
<code>salq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} \ll \text{Src}$
<code>sarq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} \gg \text{Src}$
<code>shrq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} \gg \text{Src}$
<code>xorq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} \wedge \text{Src}$
<code>andq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest} \& \text{Src}$
<code>orq</code>	<code>Src, Dest</code>	$\text{Dest} = \text{Dest}   \text{Src}$

Also called `shlq`

Arithmetic

Logical

- Watch out for argument order! *Src, Dest*  
(Warning: Intel docs use “op *Dest, Src*”)
- No distinction between signed and unsigned int (why?)

# Some Arithmetic Operations

## ■ One Operand Instructions

`incq`      `Dest`       $\text{Dest} = \text{Dest} + 1$

`decq`      `Dest`       $\text{Dest} = \text{Dest} - 1$

`negq`      `Dest`       $\text{Dest} = -\text{Dest}$

`notq`      `Dest`       $\text{Dest} = \sim\text{Dest}$

## ■ See book for more instructions



# Arithmetic Expression Example

```
long arith
(long x, long y, long z)
{
    long t1 = x+y;
    long t2 = z+t1;
    long t3 = x+4;
    long t4 = y * 48;
    long t5 = t3 + t4;
    long rval = t2 * t5;
    return rval;
}
```

```
arith:
    leaq    (%rdi,%rsi), %rax
    addq    %rdx, %rax
    leaq    (%rsi,%rsi,2), %rdx
    salq    $4, %rdx
    leaq    4(%rdi,%rdx), %rcx
    imulq   %rcx, %rax
    ret
```

## Interesting Instructions

- **leaq**: address computation
- **salq**: shift
- **imulq**: multiplication
  - But, only used once

# Understanding Arithmetic Expression

## Example

```

long arith
(long x, long y, long z)
{
    long t1 = x+y;
    long t2 = z+t1;
    long t3 = x+4;
    long t4 = y * 48;
    long t5 = t3 + t4;
    long rval = t2 * t5;
    return rval;
}

```

```

arith:
    leaq    (%rdi,%rsi), %rax    # t1
    addq    %rdx, %rax          # t2
    leaq    (%rsi,%rsi,2), %rdx
    salq    $4, %rdx           # t4
    leaq    4(%rdi,%rdx), %rcx  # t5
    imulq   %rcx, %rax         # rval
    ret

```

Register	Use(s)
%rdi	Argument <b>x</b>
%rsi	Argument <b>y</b>
%rdx	Argument <b>z</b> , <b>t4</b>
%rax	<b>t1</b> , <b>t2</b> , <b>rval</b>
%rcx	<b>t5</b>

# BTW: Intel and AT&T Syntax

## ■ Intel Syntax (aka, MASM, Microsoft Macro Assembler)

- Microsoft (Visual Studio)

## ■ AT&T Syntax (aka, GAS, Gnu ASembler)

- GNU/GCC
- Default format for objdump (GNU Binutils)

## ■ Differences

### ■ Prefixes

- \$ vs. None

Intex Syntax	AT&T Syntax
mov eax, 1	movl \$1, %eax
mov ebx, 0ffh	movl \$0xff, %ebx
int 80h	int \$0x80

### ■ Direction of Operands

Intex Syntax	AT&T Syntax
instr dest, source	instr source, dest
mov eax, [ecx]	movl (%ecx), %eax

### ■ Memory Operands

- [] vs. ()


Intex Syntax	AT&T Syntax
mov eax, [ebx]	movl (%ebx), %eax
mov eax, [ebx+3]	movl 3(%ebx), %eax

# Today: Machine Programming I: Basics

- History of Intel processors and architectures
- Assembly Basics: Registers, operands, move
- Arithmetic & logical operations
- **C, assembly, machine code**

# Turning C into Object Code

- Code in files `p1.c` `p2.c`
- Compile with command: `gcc -Og p1.c p2.c -o p`
  - Use debugging-friendly optimizations (`-Og`)
  - Put resulting binary in file `p`



# Compiling Into Assembly

## C Code (sum.c)

```
long plus(long x, long y);  
  
void sumstore(long x, long y,  
              long *dest)  
{  
    long t = plus(x, y);  
    *dest = t;  
}
```

## Generated x86-64 Assembly

```
sumstore:  
    pushq    %rbx  
    movq    %rdx, %rbx  
    call   plus  
    movq    %rax, (%rbx)  
    popq    %rbx  
    ret
```

Obtain (on x86 machine) with command

```
gcc -Og -S sum.c
```

Produces file `sum.s`

**Warning:** Will get very different results on non-x86\_64 machines (Mac OS-X, ...) due to different versions of gcc and different compiler settings.

# What it really looks like

```
        .globl sumstore
        .type  sumstore, @function
sumstore:
.LFB35:
        .cfi_startproc
pushq   %rbx
        .cfi_def_cfa_offset 16
        .cfi_offset 3, -16
movq    %rdx, %rbx
call    plus
movq    %rax, (%rbx)
popq    %rbx
        .cfi_def_cfa_offset 8
ret
        .cfi_endproc
.LFE35:
        .size  sumstore, .-sumstore
```

# What it really looks like

```

        .globl  sumstore
        .type   sumstore, @function
sumstore:
.LFB35:
        .cfi_startproc
pushq   %rbx
        .cfi_def_cfa_offset 16
        .cfi_offset 3, -16
movq    %rdx, %rbx
call    plus
movq    %rax, (%rbx)
popq    %rbx
        .cfi_def_cfa_offset 8
ret
        .cfi_endproc
.LFE35:
        .size   sumstore, .-sumstore

```

Things that look weird  
and are preceded by a ‘  
are generally directives.

```

sumstore:
    pushq   %rbx
    movq    %rdx, %rbx
    call    plus
    movq    %rax, (%rbx)
    popq    %rbx
    ret

```



# Object Code

## Code for `sumstore`

0x0400595:

0x53

0x48

0x89

0xd3

0xe8

0xf2

0xff

0xff

0xff

0x48

0x89

0x03

0x5b

0xc3

- Total of 14 bytes

- Each instruction 1, 3, or 5 bytes

- Starts at address 0x0400595

## ■ Assembler

- Translates `.s` into `.o`
- Binary encoding of each instruction
- Nearly-complete image of executable code
- Missing linkages between code in different files

## ■ Linker

- Resolves references between files
- Combines with static run-time libraries
  - E.g., code for `malloc`, `printf`
- Some libraries are *dynamically linked*
  - Linking occurs when program begins execution

# Machine Instruction Example

```
*dest = t;
```

```
movq %rax, (%rbx)
```

```
0x40059e: 48 89 03
```

## ■ C Code

- Store value `t` where designated by `dest`

## ■ Assembly

- Move 8-byte value to memory
  - Quad words in x86-64 parlance
- Operands:
  - `t`: Register `%rax`
  - `dest`: Register `%rbx`
  - `*dest`: Memory `M[%rbx]`

## ■ Object Code

- 3 bytes at address `0x40059e`
- Compact representation of the assembly instruction
- (Relatively) easy for hardware to interpret

# Disassembling Object Code

## Disassembled

```
000000000400595 <sumstore>:
 400595: 53                push   %rbx
 400596: 48 89 d3          mov    %rdx,%rbx
 400599: e8 f2 ff ff ff   callq 400590 <plus>
 40059e: 48 89 03          mov    %rax,(%rbx)
 4005a1: 5b                pop    %rbx
 4005a2: c3                retq
```

## ■ Disassembler

`objdump -d sum`

- Useful tool for examining object code
- Analyzes bit pattern of series of instructions
- Produces approximate rendition of assembly code
- Can be run on either a `.out` (complete executable) or `.o` file

# What Can be Disassembled?

```
% objdump -d WINWORD.EXE
```

```
WINWORD.EXE: file format pei-i386
```

```
No symbols in "WINWORD.EXE".
```

```
Disassembly of section .text:
```

```
30001000 <.text>:
```

```
30001000:
```

```
30001001:
```

```
30001003:
```

```
30001005:
```

```
3000100a:
```

**Reverse engineering forbidden by  
Microsoft End User License Agreement**

- Anything that can be interpreted as executable code
- Disassembler examines bytes and reconstructs assembly source

# Machine Programming I: Summary

- **History of Intel processors and architectures**
  - Evolutionary design leads to many quirks and artifacts
- **Assembly Basics: Registers, operands, move**
  - The x86-64 move instructions cover wide range of data movement forms
- **Arithmetic**
  - C compiler will figure out different instruction combinations to carry out computation
- **C, assembly, machine code**
  - New forms of visible state: program counter, registers, ...
  - Compiler must transform statements, expressions, procedures into low-level instruction sequences